
// Quelle: Julian Aichholzer, "Einführung in lineare Strukturgleichungsmodelle mit Stata" 

 

// HINWEIS: Verwendete Daten werden jeweils mit 

// "Neuer Datensatz" eingeleitet! 

 

// Zusatzpakete installieren 

ssc install relicoef 

ssc install condisc 

net install http://repec.org/bocode/e/estout/, force 

 

************************************************************************************** 

*** Neuer Datensatz, fiktive Daten für Regression – Variante 1 *** 

 

// Beispiel 1: Erstellung fiktiver Daten für Analysebeispiele – Variante 1 

 

clear  

ssd init y x1 x2 

ssd set observations 500 

ssd set means 2 3 4 

ssd set sd 5 6 7 

ssd set correlations 1 \ .4 1 \ .3 .25 1 

ssd list 

 

************************************************************************************** 

*** Neuer Datensatz, fiktive Daten für Regression – Variante 2 *** 

 

// Beispiel 2: Erstellung fiktiver Daten für Analysebeispiele – Variante 2 

 

clear  

matrix C = (1, .4, .3 \ .4, 1, .25\ .3, .25, 1)  

corr2data y x1 x2 , n(500) corr(C) means(2 3 4) sds(5 6 7)  

correlate y x1 x2, means covariance 

 

// Beispiel 3: Schritte der Effektzerlegung in der linearen Regression 

 

regress y x1, cformat(%9.3f) noheader 

regress y x1 x2, cformat(%9.3f) noheader // Simultane Schätzung 

quietly regress x1 x2 // Schritt 1 

predict u, residuals 

quietly regress y x2 // Schritt 2 

predict v, residuals 

regress v u, cformat(%9.3f) noheader // Schritt 3 

 

// Beispiel 4: Standardisierung von Regressionskoeffizienten (lineare Regression) 

 

regress y x1 x2, beta cformat(%9.3f) noheader 

egen y_z = std(y)  

egen x1_z = std(x1) 

egen x2_z = std(x2) 

regress y_z x1_z x2_z, beta cformat(%9.3f) noheader 

 

// Beispiel 5: Reproduktion der Kovarianzstruktur in der linearen Regression 

 

regress y x1 x2, cformat(%9.3f) noheader 

predict y_resid, residual 

quietly sum y_resid 

scalar psi = r(Var) 

scalar Var_y = 25 

scalar gamma1 = _b[x1] 

scalar gamma2 = _b[x2] 

scalar phi11 = 36 

scalar phi22 = 49 

scalar phi12 = 10.5 

scalar list 

display _newline "Var(y) geschätzt = " as res %4.3f /// 

(gamma1^2*phi11) + (2*gamma1*gamma2*phi12) + (gamma2^2*phi22) + psi 

display _newline "Cov(y,x2) geschätzt = " as res %4.3f /// 

(gamma1*phi12) + (gamma2*phi22) 

 

// Beispiel 6: Reproduktion der Mittelwertstruktur in der linearen Regression 

 

tabstat y x1 x2, stat(mean) 

regress y x1 x2, cformat(%9.3f) noheader 

display _newline "E(y) = " as res %4.3f /// 

_b[_cons] + _b[x1]*3 + _b[x2]*4 

 

// Beispiel 7: Vergleich lineare OLS-Regression vs. Regression mittels SEM 

 



regress y x1 x2, cformat(%4.3f)  

sem (y <- x1 x2), cformat(%4.3f) 

 

// Beispiel 8: Berechnung der erklärten Varianz – Variante 1 

 

sem (y <- x1 x2), standardized cformat(%4.3f) noheader nodescribe 

display _newline "R2 = " 1-0.797 

estat eqgof, format(%4.3f) 

display _newline "R2 = " as res %4.3f /// 

1-(19.89/24.95) 

 

// Beispiel 9: Berechnung der erklärten Varianz – Variante 2 

 

quietly sem (y <- x1 x2) 

predict y_hat, xb 

correlate y_hat y 

display _newline "R2 = " as res %4.3f /// 

r(rho)^2  

 

// Beispiel 10: Vergleich F-Test (Regression und ANOVA) vs. Wald-Test für SEM 

 

 * Neue Variable 

 egen z = cut(x2), at(-100, -1.5,1.3,1.5, 100) 

 recode z (-100=2) (-1.5=1) (1.3=2) (1.5=3) 

 tab z, gen(zdum) 

regress y zdum2 zdum3, cformat(%4.3f)  

oneway y z 

quietly sem (y <- zdum2 zdum3) 

estat eqtest 

 

// Beispiel 11: Regression ohne und mit Minderungskorrektur 

 

 * Neue Variable 

 generate x=x2 

reg y x, cformat(%4.3f) 

eivreg y x, rel(x .75) cformat(%4.3f)  

 

// Beispiel 12: Regression mit Minderungskorrekturen: SEM-Ansatz 

 

sem (Xi1 Xi2 -> Eta1) (Eta1 -> y) (Xi1 -> x1) (Xi2 -> x2), latent(Xi1 Xi2 Eta1) /// 

reliability(y .8 x1 .75 x2 .60) standardized noheader nomeans cformat(%4.3f) 

 

// Beispiel 17: Testen einzelner Modellparameter 

 

sem (y <- x1 x2), cformat(%4.3f) coeflegend noheader nodescribe 

test _b[y:x1]=_b[y:x2] 

*estat stdize: test _b[y:x1]=_b[y:x2] 

 

// Beispiel 22: Ergebnisdarstellung über Tabellen und Speichern der Ergebnisse 

 

quietly sem (y <- x1 x2), standardized 

estimates store m1 

esttab m1, main(b_std 3) not star(* .05 ** .01 *** .001) noconstant 

esttab m1, main(b_std 2) not star(* .05 ** .01) noconstant, /// 

using "C:\Temp\Erg1.rtf", replace  

 

************************************************************************************** 

*** Neuer Datensatz, KTT und Minderungskorrektur *** 

 

clear 

matrix C = (1, .8, 0, 0 \ .8, 1, 0, 0 \ 0, 0, 1, 0 \ 0, 0, 0, 1)  

quietly corr2data t1 t2 e1 e2, n(250) corr(C) means(0 0 0 0) sds(10 10 6 8)  

gen x1=t1+e1 // KTT: beobachteter Wert x=t+e 

gen x2=t2+e2 

 lab var x1 "Gemessene Testleistung (Rel = .73)" 

 lab var x2 "Gemessener Lernaufwand (Rel = .61)" 

 lab var t1 "Wahre Testleistung" 

 lab var t2 "Wahrer Lernaufwand" 

  

// Abbildung 10: Abschwächung der Korrelation durch Messfehler (fiktive Daten) 

 

sc t1 t2, scheme(s1mono) xlab(none) ylab(none) mcol(black) saving(g1.gph, replace)  

sc x1 x2, scheme(s1mono) xlab(none) ylab(none) mcol(black) saving(g2.gph, replace)  

graph combine g1.gph g2.gph, ycom xcom scheme(s1mono) 

 

// Tabelle 6: Fiktive Korrelationsmatrix (wahre Werte, Messfehler, Messungen) 

 

cor t1 t2 e1 e2 x1 x2 



mat list r(C), format(%4.3f) noheader 

* Minderungskorrektur - Rückrechnung 

sem (t1 -> x1) (t2 -> x2), /// 

latent (t1 t2) rel(x1 .734 x2 .610) stand 

display sqrt(.734)*.80*sqrt(.610)  

 

*** Neuer Datensatz, fiktive Daten für Faktorenanalyse *** 

 

clear 

matrix C =I(6) // Idenity matrix 6x6 

mat define C[1,2]=0.01 

mat define C[2,1]=0.01 

mat list C 

quietly corr2data xi1 xi2 e1 e2 e3 e4 , n(500) corr(C) means(0 0 0 0 0 0) sds(1 1 1 1 1 1)  

gen y1 = 0.55*xi1 + 0.27*xi2 + 0.90*e1 

gen y2 = 0.42*xi1 + 0.29*xi2 + 0.47*e2 

gen y3 = 0.57*xi1 + 0.22*xi2 + 0.52*e3 

gen y4 = 0.35*xi1 + 0.95*xi2 + 0.25*e4 

 

// Beispiel 13: Varianten der Spezifikation eines 1-Faktor-Messmodells (fiktive Daten) 

 

sem (Xi1 -> y1 y2 y3 y4), standardized cformat(%5.4f) nomeans noheader nodescribe 

factor y1 y2 y3 y4, ml factors(1)  

 

// Beispiel 14: Schätzung der Item-Reliabilität bzw. Kommunalität (fiktive Daten) 

 

quietly factor y1 y2 y3 y4, factors(1) ml 

matrix eins = (1,1,1,1)' 

matrix R2 = eins-e(Psi)' // Berechnung d. Item-Reliabilität nach -factor- 

matrix colnames R2 = "R2" 

matrix list R2, noheader format(%4.3f) 

quietly sem (Xi1 -> y1 y2 y3 y4) 

estat eqgof, format(%4.3f) // Berechnung d. Item-Reliabilität mit -eqgof- 

 

// Beispiel 15: Reliabilität einer Skala nach Cronbach (fiktive Daten) 

 

alpha y1 y2 y3 y4, casewise std // Alpha nach Cronbach 

display _newline "Rho(Alpha) = " as res %4.3f /// 

r(k)*r(rho) / [1+(r(k)-1)*r(rho)] // Alpha über durchschnittl. Korrelation  

 

// Beispiel 16: Varianten zur Schätzung der Composite Reliability (fiktive Daten) 

 

quietly sem (Xi1 -> y1 y2 y3 y4) 

relicoef // Composite reliability nach Raykov 

generate hv=0 // Composite reliability über latente „Phantomvariable” (C) 

quietly sem (Xi -> y1-y4) (y1-y4@1 -> C) (C -> hv@0), /// 

var(e.C@0 e.hv@0) latent(Xi C) 

quietly estat framework, fitted standardized  

matrix list r(Sigma), format(%5.4f) noheader 

dis _newline "Rho(C) = " as res %4.3f /// 

.8765^2 // Berechnung der composite reliability 

 

// Beispiel 21: Manuelle Berechnung des skalierten (Satorra-Bentler) X2-Differenztests 

 

*sem paths restricted …, vce(sbentler) 

scalar T0 = e(chi2sb_ms) 

scalar d0 = e(df_ms) 

scalar c0 = e(chi2_ms)/e(chi2sb_ms)  

*sem paths unrestricted …, vce(sbentler) 

scalar T1 = e(chi2sb_ms) 

scalar d1 = e(df_ms) 

scalar c1 = e(chi2_ms)/e(chi2sb_ms)  

scalar deltaT = (T0*c0-T1*c1)*(d0-d1) / (c0*d0-c1*d1) 

scalar list 

display _newline "Skal. Chi2-Diff. = "deltaT /// 

_newline "Diff. d.f. = "d0-d1 /// 

_newline "p = "as res %5.4f chi2tail(d0-d1, deltaT) 

 

************************************************************************************** 

*** Neuer Datensatz, AUTNES-Daten *** 

 

use "ZA5859_de_v2-0-0.dta", clear // Laden des Datensatzes (aus dem working directory) 

 

// Beispiel 23: Aufbereitung der Originaldaten und Variablen (AUTNES-Daten) 

 

* Laden des Datensatzes (aus dem working directory) 

use "ZA5859_de_v2-0-0.dta", clear 

 



* Originale Namen der Items umbenennen 

rename (w1_q22) (euint) 

rename (w1_q82x2 w1_q82x4 w1_q82x5) (rwa1 rwa2 rwa3) 

rename (w1_q26x10 w1_q26x11 w1_q26x12 w1_q76x1) (imm1 imm2 imm3 imm4) 

rename (w1_q83x5 w1_q83x10) (ope1 ope2) 

 

* Ausschluss von Missing Values: 88 (weiß nicht) und 99 (verweigert) 

mvdecode euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2, mv(88 99) 

 

* Achtung! Semantische Recodierung: hohe Werte = hohe Merkmalswerte 

foreach var of varlist imm2 imm3 imm4 ope2 { 

 recode `var' (5=1) (4=2) (3=3) (2=4) (1=5) 

 } 

  

// Beispiel 24: Verwendete Variablenliste und deskriptive Statistiken (AUTNES-Daten) 

 

describe euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2  

summarize euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2 

 

// Beispiel 25: Item-Korrelationen (AUTNES-Daten) 

 

quietly correlate euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2 

matrix list r(C), format(%4.2f) noheader 

 

// Beispiel 26: Prüfung der Normalverteilung der Daten (AUTNES-Daten) 

 

sfrancia euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2 

mvtest normality euint rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2 

 

// Beispiel 27: Beispiel für eine explorative Faktorenanalyse (EFA) 

 

quietly factor rwa1 rwa2 rwa3 imm1 imm2 imm3 imm4 ope1 ope2, factors(3) ml 

rotate, oblique quartimin blank(.15) 

 

// Beispiel 28: Postestimation-Befehl in der EFA: KMO-Kriterium 

 

estat kmo 

 

// Beispiel 29: Korrelationsmatrix der Faktoren (EFA) 

 

estat common, format(%6.3f) 

 

// Beispiel 30: Beispiel für eine konfirmatorische Faktorenanalyse (CFA) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) 

estat framework, standardized format(%6.3f) compact 

estat eqgof, format(%4.3f) 

 

// Beispiel 31: Gütemaße des 3-Faktoren-Modells (CFA)  

 

estat gof, stats(all) 

 

// Beispiel 32: Likelihood-Ratio-Test (?2-Differenztest) für ein 2- vs. 3-Faktoren-Modell 

 

eststo mf3: quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) 

eststo mf2: quietly sem (IMMRWA -> imm? rwa?) (OPENN -> ope?) 

lrtest mf3 mf2 

 

// Beispiel 33: Konvergente und diskriminante Validität von Items (AUTNES-Daten) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) 

condisc 

 

// Beispiel 34: Reliabilitätsschätzung über Alpha nach Cronbach (AUTNES-Daten) 

 

alpha ope?, casewise std  

alpha rwa?, casewise std 

alpha imm?, casewise std 

 

// Beispiel 35: Reliabilitätsschätzung über composite reliability (AUTNES-Daten)  

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) 

relicoef // Reliabilitätsschätzung mittels relicoef 

generate hv=0 // Reliabilitätsschätzung mit latenter Phantomvariable (Bsp. IMM) 

quietly sem (IMM -> imm?) (imm?@1 -> C) (C -> hv@0), var(e.C@0 e.hv@0) 

quietly estat framework, fitted standardized 

matrix list r(Sigma), format(%5.4f) noheader 

dis _newline "Rho(C) = " as res %6.3f /// 



.9234^2 

 

// Beispiel 36: Bildung von Summenindizes (AUTNES-Daten) 

 

generate c_openn = ope1+ope2 

generate c_rwa = rwa1+rwa2+rwa3  

generate c_imm = imm1+imm2+imm3+imm4  

 

// Beispiel 37: Korrelationsmatrix der untersuchten Variablen (Summenscores) 

 

correlate euint c_openn c_rwa c_imm 

 

// Beispiel 38: Korrelationsmatrix der untersuchten Variablen (SEM-basiert) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) (euint)  

estat framework, standardized format(%9.3f) 

 

// Beispiel 39: Klassische lineare OLS-Regression (AUTNES-Daten) 

 

regress euint c_openn c_rwa c_imm, beta cformat(%4.3f) 

 

// Beispiel 40: Berechnung standardisierter totaler/indirekter Effekte (SB-Schätzer) 

 

sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) /// 

(OPENN -> RWA) (RWA -> IMM) (IMM -> euint), coeflegend vce(sbentler) 

estat stdize: nlcom _b[euint:IMM]*_b[IMM:RWA]*_b[RWA:OPENN], noheader // (OPENN -> euint) 

estat stdize: nlcom _b[euint:IMM]*_b[IMM:RWA], noheader // (RWA -> euint) 

estat stdize: nlcom _b[IMM:RWA]*_b[RWA:OPENN], noheader // (OPENN -> RWA) 

 

// Beispiel 41: Modellvergleich mit Satorra-Bentler-?2-Differenztest (AUTNES-Daten) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) /// 

(OPENN -> RWA) (RWA -> IMM) (IMM -> euint), vce(sbentler) 

 scalar T0 = e(chi2sb_ms) 

 scalar d0 = e(df_ms) 

 scalar c0 = e(chi2_ms)/e(chi2sb_ms) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) /// 

(OPENN -> RWA) (OPENN RWA -> IMM) (RWA IMM -> euint), vce(sbentler) 

 scalar T1 = e(chi2sb_ms) 

 scalar d1 = e(df_ms) 

 scalar c1 = e(chi2_ms)/e(chi2sb_ms)  

scalar deltaT = (T0*c0-T1*c1)*(d0-d1) / (c0*d0-c1*d1) 

scalar list 

display _newline "Skal. Chi2-Diff. = "deltaT /// 

_newline "Diff. d.f. = "d0-d1 /// 

_newline "p = "as res %5.4f chi2tail(d0-d1, deltaT) 

 

// Beispiel 42: Auszug der Modifikationsindizes (AUTNES-Daten) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) /// 

(OPENN -> RWA) (RWA -> IMM) (IMM -> euint) 

estat mi 

 

// Beispiel 43: Standardisierte Residuen des Modells (AUTNES-Daten) 

 

quietly sem (OPENN -> ope?) (RWA -> rwa?) (IMM -> imm?) /// 

(OPENN -> RWA) (RWA -> IMM) (IMM -> euint), vce(sbentler) 

quietly estat residuals, standardized 

matrix list r(sres_cov), format(%4.2f) noheader 

